Rules for integrands of the form $(f x)^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n$

1. $\int (f x)^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d$

   1. $\int (f x)^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0$

      1. $\int x (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0$

         1: $\int \frac{x (a + b \text{ArcSinh}[c x])^n}{d + e x^2} \, dx$ when $e == c^2 d \land n \in \mathbb{Z}^+$

- Derivation: Integration by substitution

  Basis: If $e == c^2 d$, then $\frac{x}{d+e x^2} == \frac{1}{e} \text{Subst}[\text{Tanh}[x], x, \text{ArcSinh}[c x]] \, \partial_x \text{ArcSinh}[c x]$

- Note: If $n \in \mathbb{Z}^+$, then $(a + b x)^n \text{Tanh}[x]$ is integrable in closed-form.

- Rule: If $e == c^2 d \land n \in \mathbb{Z}^+$, then

$$\int \frac{x (a + b \text{ArcSinh}[c x])^n}{d + e x^2} \, dx \longrightarrow \frac{1}{e} \text{Subst}\left[\int (a + b x)^n \text{Tanh}[x] \, dx, x, \text{ArcSinh}[c x]\right]$$

- Program code:

```
Int[x_*(a_.+b_.*ArcSinh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/e*Subst[Int[(a+b*x)^n*Tanh[x],x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

2: $\int x\left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n>0 \land p \neq -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $x\left(d+e x^2\right)^p == \partial_x \frac{\left(d+e x^2\right)^{p+1}}{2 e (p+1)}$

Basis: $\partial_x (a+b\,\text{ArcSinh}[c\,x])^n == \frac{b\,c\,n\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{\left(d+e x^2\right)^p}{\left(1+c^2 x^2\right)^p} == 0$

Rule: If $e == c^2 d \land n>0 \land p \neq -1$, then

$$\int x\left(d+e x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

$$\longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^n}{2 e (p+1)}-\frac{b\,c\,n}{2 e (p+1)} \int \frac{\left(d+e x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2 x^2}}\,dx$$

$$\longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^n}{2 e (p+1)}-\frac{b\,n\left(d+e x^2\right)^p}{2 c (p+1)\left(1+c^2 x^2\right)^p} \int \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  b*n/(2*c*(p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && NeQ[p,-1]
```

2. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land m+2p+3 == 0$

1: $\int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{x \left(d+e x^2\right)} \, dx$ when $e == c^2 d \land n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $e == c^2 d$, then $\frac{1}{x \left(d+e x^2\right)} == \frac{1}{d} \operatorname{Subst}\left[\frac{1}{\operatorname{Cosh}[x] \operatorname{Sinh}[x]}, x, \operatorname{ArcSinh}[c x]\right] \partial_x \operatorname{ArcSinh}[c x]$

Rule: If $e == c^2 d \land n \in \mathbb{Z}^+$, then

$$\int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{x \left(d+e x^2\right)} \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \frac{(a+b x)^n}{\operatorname{Cosh}[x] \operatorname{Sinh}[x]} \, dx, x, \operatorname{ArcSinh}[c x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  1/d*Subst[Int[(a+b*x)^n/(Cosh[x]*Sinh[x]),x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land m+2p+3 == 0 \land m \neq -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: If $m+2p+3 == 0$, then $(f x)^m \left(d+e x^2\right)^p == \partial_x \frac{(f x)^{m+1} \left(d+e x^2\right)^{p+1}}{d f (m+1)}$

Basis: $\partial_x (a+b \operatorname{ArcSinh}[c x])^n == \frac{b c n (a+b \operatorname{ArcSinh}[c x])^{n-1}}{\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{\left(d+e x^2\right)^p}{\left(1+c^2 x^2\right)^p} == 0$

Rule: If $e == c^2 d \land n > 0 \land m+2p+3 == 0 \land m \neq -1$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

$$\longrightarrow\ \frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^n}{d\,f\,(m+1)} - \frac{b\,c\,n}{d\,f\,(m+1} \int \frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1+c^2\,x^2}}\,dx$$

$$\longrightarrow\ \frac{(f\,x)^{m+1} \left(d+e\,x^2\right)^{p+1} (a+b\,\text{ArcSinh}[c\,x])^n}{d\,f\,(m+1)} - \frac{b\,c\,n \left(d+e\,x^2\right)^p}{f\,(m+1) \left(1+c^2\,x^2\right)^p} \int (f\,x)^{m+1} \left(1+c^2\,x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1]
```

3. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,p>0$

1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2\,d\,\wedge\,p>0$

1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+$

1. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+\wedge\,\frac{m-1}{2}\in\mathbb{Z}^-$

1: $\int \frac{(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])}{x}\,dx$ when $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+$

Derivation: Inverted integration by parts

Rule: If $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+$, then

$$\int \frac{(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])}{x}\,dx \longrightarrow$$
$$\frac{(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])}{2\,p}-\frac{b\,c\,d^p}{2\,p}\int (1+c^2\,x^2)^{p-\frac{1}{2}}\,dx+d\int \frac{(d+e\,x^2)^{p-1}\,(a+b\,\text{ArcSinh}[c\,x])}{x}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])/x_,x_Symbol] :=
  (d+e*x^2)^p*(a+b*ArcSinh[c*x])/(2*p) -
  b*c*d^p/(2*p)*Int[(1+c^2*x^2)^(p-1/2),x] +
  d*Int[(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])/x,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])\,dx$ when $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+\wedge\,\frac{m+1}{2}\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $e == c^2\,d\,\wedge\,p\in\mathbb{Z}^+\wedge\,\frac{m+1}{2}\in\mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])\, dx \longrightarrow$$
$$\frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])}{f (m+1)} -$$
$$\frac{b c d^p}{f (m+1)} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} dx - \frac{2 e p}{f^2 (m+1)} \int (f x)^{m+2} \left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcSinh}[c x])\, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])/(f*(m+1)) -
  b*c*d^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2),x] -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && IGtQ[p,0] && ILtQ[(m+1)/2,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])\, dx$ when $e = c^2 d \wedge p \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $e = c^2 d \wedge p \in \mathbb{Z}^+$, let $u = \int (f x)^m \left(d+e x^2\right)^p dx$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])\, dx \longrightarrow u\, (a+b \operatorname{ArcSinh}[c x]) - b c \int \frac{u}{\sqrt{1+c^2 x^2}}\, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

2: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])\, dx$ when $e == c^2 d \land p-\frac{1}{2} \in \mathbb{Z} \land p \neq -\frac{1}{2} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x (a+b \operatorname{ArcSinh}[c x]) == \frac{b c}{\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{d+e x^2}}{\sqrt{1+c^2 x^2}} == 0$

Note: If $p-\frac{1}{2} \in \mathbb{Z} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$, then $\int x^m \left(d+e x^2\right)^p dx$ is an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $e == c^2 d \land p-\frac{1}{2} \in \mathbb{Z} \land p \neq -\frac{1}{2} \land \left(\frac{m+1}{2} \in \mathbb{Z}^+ \lor \frac{m+2p+3}{2} \in \mathbb{Z}^-\right)$, let $u = \int x^m \left(d+e x^2\right)^p dx$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])\, dx \longrightarrow u\,(a+b \operatorname{ArcSinh}[c x]) - b c \int \frac{u}{\sqrt{1+c^2 x^2}}\, dx \longrightarrow u\,(a+b \operatorname{ArcCosh}[c x]) - \frac{b c \sqrt{d+e x^2}}{\sqrt{1+c^2 x^2}} \int \frac{u}{\sqrt{d+e x^2}}\, dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u] -
  b*c*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[SimplifyIntegrand[u/Sqrt[d+e*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegerQ[p-1/2] && NeQ[p,-1/2] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0])
```

2. $\int (f x)^m \sqrt{d+e x^2}\, (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0$

 1: $\int (f x)^m \sqrt{d+e x^2}\, (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land m < -1$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e == c^2 d \land n > 0 \land m < -1$, then

$$\int (f x)^m \sqrt{d+e x^2}\, (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \sqrt{d+e x^2}\, (a+b \operatorname{ArcSinh}[c x])^n}{f (m+1)} -$$

$$\frac{b c n \sqrt{d+e x^2}}{f (m+1) \sqrt{1+c^2 x^2}} \int (f x)^{m+1} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx - \frac{c^2 \sqrt{d+e x^2}}{f^2 (m+1) \sqrt{1+c^2 x^2}} \int \frac{(f x)^{m+2} (a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(f*(m+1)) -
  b*c*n/(f*(m+1))*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[(f*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1),x] -
  c^2/(f^2*(m+1))*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[(f*x)^(m+2)*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[m,-1]
```

2: $\int (f x)^m \sqrt{d+e x^2}\, (a+b\,\text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n \in \mathbb{Z}^+ \land (m+2 \in \mathbb{Z}^+ \lor n == 1)$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $e == c^2 d \land n \in \mathbb{Z}^+ \land (m+2 \in \mathbb{Z}^+ \lor n == 1)$, then

$$\int (f x)^m \sqrt{d+e x^2}\, (a+b\,\text{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \sqrt{d+e x^2}\, (a+b\,\text{ArcSinh}[c x])^n}{f\,(m+2)} - \frac{b c n \sqrt{d+e x^2}}{f\,(m+2) \sqrt{1+c^2 x^2}} \int (f x)^{m+1} (a+b\,\text{ArcSinh}[c x])^{n-1} \, dx + \frac{\sqrt{d+e x^2}}{(m+2) \sqrt{1+c^2 x^2}} \int \frac{(f x)^m (a+b\,\text{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} \, dx$$

–

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(f*(m+2)) -
  b*c*n/(f*(m+2))*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[(f*x)^(m+1)*(a+b*ArcSinh[c*x])^(n-1),x] +
  1/(m+2)*Simp[Sqrt[d+e*x^2]/Sqrt[1+c^2*x^2]]*Int[(f*x)^m*(a+b*ArcSinh[c*x])^n/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && IGtQ[n,0] && (IGtQ[m,-2] || EqQ[n,1])
```

3. $\int (f x)^m (d+e x^2)^p (a+b\,\text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0$

1: $\int (f x)^m (d+e x^2)^p (a+b\,\text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p > 0 \land m < -1$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n > 0 \land p > 0 \land m < -1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$
$$\frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n}{f (m+1)} -$$
$$\frac{2 e p}{f^2 (m+1)} \int (f x)^{m+2} \left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcSinh}[c x])^n \, dx - \frac{b c n \left(d+e x^2\right)^p}{f (m+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+1)) -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n>0 \land p>0 \land m \nless -1$

Derivation: Inverted integration by parts

Rule: If $e == c^2 d \land n>0 \land p>0 \land m \nless -1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n}{f (m+2 p+1)} +$$

$$\frac{2 d p}{m+2 p+1} \int (f x)^m \left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcSinh}[c x])^n \, dx - \frac{b c n \left(d+e x^2\right)^p}{f (m+2 p+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n/(f*(m+2*p+1)) +
  2*d*p/(m+2*p+1)*Int[(f*x)^m*(d+e*x^2)^(p-1)*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n/(f*(m+2*p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]]
```

4: $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx$ when $e == c^2 d \land n>0 \land m+1 \in \mathbb{Z}^-$

Rule: If $e == c^2 d \land n>0 \land m+1 \in \mathbb{Z}^-$, then

$$\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx \longrightarrow$$

$$\frac{(f x)^{m+1} (d+e x^2)^{p+1} (a+b \text{ArcSinh}[c x])^n}{d f (m+1)} -$$

$$\frac{c^2 (m+2p+3)}{f^2 (m+1)} \int (f x)^{m+2} (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx - \frac{b c n (d+e x^2)^p}{f (m+1) (1+c^2 x^2)^p} \int (f x)^{m+1} (1+c^2 x^2)^{p+\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n-1} dx$$

Programcode:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(d*f*(m+1)) -
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && ILtQ[m,-1]
```

5. $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx$ when $e == c^2 d \land n>0 \land p<-1 \land m \in \mathbb{Z}$

1: $\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx$ when $e == c^2 d \land n>0 \land p<-1 \land m-1 \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $x (d+e x^2)^p == \partial_x \frac{(d+e x^2)^{p+1}}{2 e (p+1)}$

Rule: If $e == c^2 d \land n>0 \land p<-1 \land m-1 \in \mathbb{Z}^+$, then

$$\int (f x)^m (d+e x^2)^p (a+b \text{ArcSinh}[c x])^n dx \longrightarrow$$

$$\frac{f (f x)^{m-1} (d+e x^2)^{p+1} (a+b \text{ArcSinh}[c x])^n}{2 e (p+1)} -$$

$$\frac{f^2 (m-1)}{2 e (p+1)} \int (f x)^{m-2} \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcSinh}[c x])^n \, dx - \frac{b f n \left(d+e x^2\right)^p}{2 c (p+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m-1} \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n/(2*c*(p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && IGtQ[m,1]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land n > 0 \land p < -1 \land m \in \mathbb{Z}^-$

Rule: If $e == c^2 d \land n > 0 \land p < -1 \land m \in \mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow$$

$$-\frac{(f x)^{m+1} \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcSinh}[c x])^n}{2 d f (p+1)} +$$

$$\frac{m+2 p+3}{2 d (p+1)} \int (f x)^m \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcSinh}[c x])^n \, dx + \frac{b c n \left(d+e x^2\right)^p}{2 f (p+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b \operatorname{ArcSinh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(2*d*f*(p+1)) +
  (m+2*p+3)/(2*d*(p+1))*Int[(f*x)^m*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n,x] +
  b*c*n/(2*f*(p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && (IntegerQ[m] || IntegerQ[p] || EqQ[n,1])
```

6: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d\,\wedge\,n>0\,\wedge\,m-1\in\mathbb{Z}^+\,\wedge\,m+2\,p+1\neq 0$

Rule: If $e == c^2\,d\,\wedge\,n>0\,\wedge\,m-1\in\mathbb{Z}^+\,\wedge\,m+2\,p+1\neq 0$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$
$$\frac{f\,(f\,x)^{m-1}\,(d+e\,x^2)^{p+1}\,(a+b\,\text{ArcSinh}[c\,x])^n}{e\,(m+2\,p+1)}-$$
$$\frac{f^2\,(m-1)}{c^2\,(m+2\,p+1)}\int (f\,x)^{m-2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx-\frac{b\,f\,n\,(d+e\,x^2)^p}{c\,(m+2\,p+1)\,(1+c^2\,x^2)^p}\int (f\,x)^{m-1}\,(1+c^2\,x^2)^{p+\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])^n/(e*(m+2*p+1)) -
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] -
  b*f*n/(c*(m+2*p+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[e,c^2*d] && GtQ[n,0] && IGtQ[m,1] && NeQ[m+2*p+1,0]
```

2. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n < -1$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n < -1 \land m+2\,p+1 == 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $e == c^2\,d \land m+2\,p+1 == 0$, then $\partial_x\left((f\,x)^m\,\sqrt{1+c^2\,x^2}\,(d+e\,x^2)^p\right) == \frac{f\,m\,(f\,x)^{m-1}\,(d+e\,x^2)^p}{\sqrt{1+c^2\,x^2}}$

Basis: If $e == c^2\,d$, then $\partial_x \frac{(d+e\,x^2)^p}{(1+c^2\,x^2)^p} == 0$

Rule: If $e == c^2\,d \land n < -1 \land m+2\,p+1 == 0$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \to$$

$$\frac{(f\,x)^m\,\sqrt{1+c^2\,x^2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)} - \frac{f\,m\,(d+e\,x^2)^p}{b\,c\,(n+1)\,(1+c^2\,x^2)^p}\int (f\,x)^{m-1}\,(1+c^2\,x^2)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && EqQ[m+2*p+1,0]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2\,d \land n < -1 \land 2\,p \in \mathbb{Z}^+ \land m+2\,p+1 \neq 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $e == c^2 d$, then $\partial_x \left((f\,x)^m \sqrt{1+c^2 x^2}\,(d+e\,x^2)^p\right) == \frac{f\,m\,(f\,x)^{m-1}\,(d+e\,x^2)^p}{\sqrt{1+c^2 x^2}} + \frac{c^2\,(m+2\,p+1)\,(f\,x)^{m+1}\,(d+e\,x^2)^p}{f\,\sqrt{1+c^2 x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{(d+e\,x^2)^p}{(1+c^2 x^2)^p} == 0$

- Rule: If $e == c^2 d \land n < -1 \land 2\,p \in \mathbb{Z}^+ \land m+2\,p+1 \neq 0$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^m \sqrt{1+c^2 x^2}\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)} -$$
$$\frac{f\,m\,(d+e\,x^2)^p}{b\,c\,(n+1)\,(1+c^2 x^2)^p} \int (f\,x)^{m-1}\,(1+c^2 x^2)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx -$$
$$\frac{c\,(m+2\,p+1)\,(d+e\,x^2)^p}{b\,f\,(n+1)\,(1+c^2 x^2)^p} \int (f\,x)^{m+1}\,(1+c^2 x^2)^{p-\frac{1}{2}}\,(a+b\,\text{ArcSinh}[c\,x])^{n+1}\,dx$$

- Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c^2*x^2]*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] -
  c*(m+2*p+1)/(b*f*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && LtQ[n,-1] && IGtQ[2*p,0] && NeQ[m+2*p+1,0] && IGtQ[m,-3]
```

3: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e == c^2 d \land n < -1 \land 2\,p \in \mathbb{Z} \land p \neq -\frac{1}{2}$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $e == c^2 d$, then
$$\partial_x \left((f x)^m \sqrt{1+c^2 x^2}\,\left(d+e x^2\right)^p\right) == f m (f x)^{m-1} \sqrt{1+c^2 x^2}\,\left(d+e x^2\right)^p + \frac{c^2 (2 p+1) (f x)^{m+1} \left(d+e x^2\right)^p}{f \sqrt{1+c^2 x^2}}$$

Basis: If $e == c^2 d$, then $\partial_x \frac{\left(d+e x^2\right)^p}{\left(1+c^2 x^2\right)^p} == 0$

Rule: If $e == c^2 d \land n < -1 \land 2 p \in \mathbb{Z} \land p \neq -\frac{1}{2}$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcSinh}[c x])^n \, dx$$

$$\rightarrow \frac{(f x)^m \sqrt{1+c^2 x^2}\,\left(d+e x^2\right)^p (a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)} - \frac{f m}{b c (n+1)} \int (f x)^{m-1} \sqrt{1+c^2 x^2}\,\left(d+e x^2\right)^p (a+b \text{ArcSinh}[c x])^{n+1} \, dx - \frac{c (2 p+1)}{b f (n+1)} \int \frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \text{ArcSinh}[c x])^{n+1}}{\sqrt{1+c^2 x^2}} \, dx$$

$$\rightarrow \frac{(f x)^m \sqrt{1+c^2 x^2}\,\left(d+e x^2\right)^p (a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)} - \frac{f m \left(d+e x^2\right)^p}{b c (n+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m-1} \left(1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n+1} \, dx - \frac{c (2 p+1) \left(d+e x^2\right)^p}{b f (n+1) \left(1+c^2 x^2\right)^p} \int (f x)^{m+1} \left(1+c^2 x^2\right)^{p-\frac{1}{2}} (a+b \text{ArcSinh}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Simp[Sqrt[1+c^2*x^2]*(d+e*x^2)^p]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m-1)*(1+c^2*x^2)^(p+1/2)*(a+b*ArcSinh[c*x])^(n+1),x] -
  c*(2*p+1)/(b*f*(n+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*Int[(f*x)^(m+1)*(1+c^2*x^2)^(p-1/2)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[e,c^2*d] && LtQ[n,-1] && IntegerQ[2*p] && NeQ[p,-1/2] && IGtQ[m,-3] *)
```

3. $\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx$ when $e == c^2 d$

1. $\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx$ when $e == c^2 d \land n > 0$

1: $\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx$ when $e == c^2 d \land n > 0 \land m - 1 \in \mathbb{Z}^+$

Rule: If $e == c^2 d \land n > 0 \land m - 1 \in \mathbb{Z}^+$, then

$$\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \sqrt{d + e x^2} (a + b \text{ArcSinh}[c x])^n}{e m} -$$

$$\frac{b f n \sqrt{1 + c^2 x^2}}{c m \sqrt{d + e x^2}} \int (f x)^{m-1} (a + b \text{ArcSinh}[c x])^{n-1} \, dx - \frac{f^2 (m-1)}{c^2 m} \int \frac{(f x)^{m-2} (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcSinh[c*x])^n/(e*m) -
  b*f*n/(c*m)*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Int[(f*x)^(m-1)*(a+b*ArcSinh[c*x])^(n-1),x] -
  f^2*(m-1)/(c^2*m)*Int[((f*x)^(m-2)*(a+b*ArcSinh[c*x])^n)/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[n,0] && IGtQ[m,1]
```

2: $\int \frac{x^m (a+b \text{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx$ when $e == c^2 d \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Basis: If $m \in \mathbb{Z}$, then $\frac{x^m}{\sqrt{1+c^2 x^2}} == \frac{1}{c^{m+1}}$ Subst[Sinh[x]^m, x, ArcSinh[c x]] ∂_xArcSinh[c x]

Note: If $n \in \mathbb{Z}^+$, then $(a+b x)^n \text{Sinh}[x]$ is integrable in closed-form.

Rule: If $e == c^2 d \wedge n \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}$, then

$$\int \frac{x^m (a+b \text{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \, dx \rightarrow \frac{\sqrt{1+c^2 x^2}}{c^{m+1} \sqrt{d+e x^2}} \text{Subst}\left[\int (a+b x)^n \text{Sinh}[x]^m \, dx, x, \text{ArcSinh}[c x]\right]$$

Program code:

```
Int[x_^m_*(a_.+b_.*ArcSinh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/c^(m+1)*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Subst[Int[(a+b*x)^n*Sinh[x]^m,x],x,ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[n,0] && IntegerQ[m]
```

3: $\int \frac{(f x)^m (a+b \text{ArcSinh}[c x])}{\sqrt{d+e x^2}} \, dx$ when $e == c^2 d \wedge m \notin \mathbb{Z}$

Rule: If $e == c^2 d \wedge m \notin \mathbb{Z}$, then

$$\int \frac{(f x)^m (a+b \text{ArcSinh}[c x])}{\sqrt{d+e x^2}} \, dx \rightarrow$$

$$\frac{(f x)^{m+1} \sqrt{1+c^2 x^2} (a+b \text{ArcSinh}[c x])}{f (m+1) \sqrt{d+e x^2}} \text{Hypergeometric2F1}\left[\frac{1}{2}, \frac{1+m}{2}, \frac{3+m}{2}, -c^2 x^2\right] -$$

$$\frac{b\,c\,(f\,x)^{m+2}\sqrt{1+c^2\,x^2}}{f^2\,(m+1)\,(m+2)\,\sqrt{d+e\,x^2}}\,\text{HypergeometricPFQ}\left[\left\{1,\,1+\frac{m}{2},\,1+\frac{m}{2}\right\},\,\left\{\frac{3}{2}+\frac{m}{2},\,2+\frac{m}{2}\right\},\,-c^2\,x^2\right]$$

–

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcSinh[c_.*x_])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*(a+b*ArcSinh[c*x])*
    Hypergeometric2F1[1/2,(1+m)/2,(3+m)/2,-c^2*x^2] -
  b*c*(f*x)^(m+2)/(f^2*(m+1)*(m+2))*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*
    HypergeometricPFQ[{1,1+m/2,1+m/2},{3/2+m/2,2+m/2},-c^2*x^2] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && Not[IntegerQ[m]]
```

2: $\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx$ when $e == c^2 d \land n < -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b \text{ArcSinh}[c x])^n}{\sqrt{1+c^2 x^2}} == \partial_x \frac{(a+b \text{ArcSinh}[c x])^{n+1}}{b c (n+1)}$

Basis: If $e == c^2 d$, then $\partial_x \frac{(f x)^m \sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == \frac{f m (f x)^{m-1} \sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}}$

Basis: If $e == c^2 d$, then $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $e == c^2 d \land n < -1$, then

$$\int \frac{(f x)^m (a + b \text{ArcSinh}[c x])^n}{\sqrt{d + e x^2}} \, dx \longrightarrow$$

$$\frac{(f x)^m \sqrt{1 + c^2 x^2} (a + b \text{ArcSinh}[c x])^{n+1}}{b c (n + 1) \sqrt{d + e x^2}} - \frac{f m \sqrt{1 + c^2 x^2}}{b c (n + 1) \sqrt{d + e x^2}} \int (f x)^{m-1} (a + b \text{ArcSinh}[c x])^{n+1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcSinh[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^m/(b*c*(n+1))*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*(a+b*ArcSinh[c*x])^(n+1) -
  f*m/(b*c*(n+1))*Simp[Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]]*Int[(f*x)^(m-1)*(a+b*ArcSinh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && LtQ[n,-1]
```

4: $\int x^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n \, dx$ when $e == c^2 d \land 2 p + 2 \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $e == c^2 d$, then $\partial_x \frac{(d+e x^2)^p}{(1+c^2 x^2)^p} == 0$

Basis: If $m \in \mathbb{Z}$, then
$x^m (1 + c^2 x^2)^p ==$
$\frac{1}{b c^{m+1}} \text{Subst}\left[\text{Sinh}\left[-\frac{a}{b} + \frac{x}{b}\right]^m \text{Cosh}\left[-\frac{a}{b} + \frac{x}{b}\right]^{2p+1}, x, a + b \text{ArcSinh}[c x]\right] \partial_x (a + b \text{ArcSinh}[c x])$

Note: If $2p + 2 \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+$, then $x^n \text{Sinh}[-\frac{a}{b} + \frac{x}{b}]^m \text{Cosh}[-\frac{a}{b} + \frac{x}{b}]^{2p+1}$ is integrable in closed-form.

Rule: If $e == c^2 d \land 2p + 2 \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+$, then

$$\int x^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n dx$$

$$\rightarrow \frac{(d + e x^2)^p}{(1 + c^2 x^2)^p} \int x^m (1 + c^2 x^2)^p (a + b \text{ArcSinh}[c x])^n dx$$

$$\rightarrow \frac{(d + e x^2)^p}{b c^{m+1} (1 + c^2 x^2)^p} \text{Subst}\left[\int x^n \text{Sinh}\left[-\frac{a}{b} + \frac{x}{b}\right]^m \text{Cosh}\left[-\frac{a}{b} + \frac{x}{b}\right]^{2p+1} dx, x, a + b \text{ArcSinh}[c x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c^(m+1))*Simp[(d+e*x^2)^p/(1+c^2*x^2)^p]*
    Subst[Int[x^n*Sinh[-a/b+x/b]^m*Cosh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcSinh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[e,c^2*d] && IGtQ[2*p+2,0] && IGtQ[m,0]
```

5: $\int (f x)^m (d + e x^2)^p (a + b \text{ArcSinh}[c x])^n dx$ when $e == c^2 d \land p + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{m+1}{2} \notin \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $e == c^2 d \land p + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{m+1}{2} \notin \mathbb{Z}^+$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx \longrightarrow \int \frac{(a+b \operatorname{ArcSinh}[c x])^n}{\sqrt{d+e x^2}} \operatorname{ExpandIntegrand}\left[(f x)^m \left(d+e x^2\right)^{p+\frac{1}{2}}, x\right] dx$$

- **Program code:**

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n/Sqrt[d+e*x^2],(f*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[e,c^2*d] && IGtQ[p+1/2,0] && Not[IGtQ[(m+1)/2,0]] && (EqQ[m,-1] || EqQ[m,-2])
```

2. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x])^n \, dx$ when $e \neq c^2 d$

   1: $\int x \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx$ when $e \neq c^2 d \land p \neq -1$

- **Derivation: Integration by parts**

- **Basis:: If** $p \neq -1$, **then** $x \left(d+e x^2\right)^p = \partial_x \frac{\left(d+e x^2\right)^{p+1}}{2 e (p+1)}$

- **Rule: If** $e \neq c^2 d \land p \neq -1$, **then**

$$\int x \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx \longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcSinh}[c x])}{2 e (p+1)} - \frac{b c}{2 e (p+1)} \int \frac{\left(d+e x^2\right)^{p+1}}{\sqrt{1+c^2 x^2}} \, dx$$

- **Program code:**

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcSinh[c*x])/(2*e*(p+1)) - b*c/(2*e*(p+1))*Int[(d+e*x^2)^(p+1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[e,c^2*d] && NeQ[p,-1]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx$ when $e \neq c^2 d \land p \in \mathbb{Z} \land \left(p>0 \lor \frac{m-1}{2} \in \mathbb{Z}^+ \land m+p \leq 0\right)$

Derivation: Integration by parts

Note: If $\frac{m-1}{2} \in \mathbb{Z}^+ \land p \in \mathbb{Z}^- \land m+p \geq 0$, then $\int (f x)^m \left(d+e x^2\right)^p$ is a rational function.

Rule: If $e \neq c^2 d \land p \in \mathbb{Z} \land \left(p>0 \lor \frac{m-1}{2} \in \mathbb{Z}^+ \land m+p \leq 0\right)$, let $u=\int (f x)^m \left(d+e x^2\right)^p dx$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcSinh}[c x]) \, dx \longrightarrow u \, (a+b \operatorname{ArcSinh}[c x]) - b c \int \frac{u}{\sqrt{1+c^2 x^2}} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcSinh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/Sqrt[1+c^2*x^2],x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[e,c^2*d] && IntegerQ[p] && (GtQ[p,0] || IGtQ[(m-1)/2,0] && LeQ[m+p,0])
```

3: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$ when $e\neq c^2 d\ \land\ n\in\mathbb{Z}^+\land\ p\in\mathbb{Z}\ \land\ m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $e\neq c^2 d\ \land\ n\in\mathbb{Z}^+\land\ p\in\mathbb{Z}\ \land\ m\in\mathbb{Z}$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \to \int (a+b\,\text{ArcSinh}[c\,x])^n\,\text{ExpandIntegrand}\left[(f\,x)^m \left(d+e\,x^2\right)^p,\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcSinh[c*x])^n,(f*x)^m*(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[e,c^2*d] && IGtQ[n,0] && IntegerQ[p] && IntegerQ[m]
```

U: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$

Rule:

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx \to \int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^p*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

Rules for integrands of the form $(h x)^m (d + e x)^p (f + g x)^q (a + b \text{ArcSinh}[c x])^n$

1: $\int (h x)^m (d + e x)^p (f + g x)^q (a + b \text{ArcSinh}[c x])^n \, dx$ when $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land d > 0 \land \frac{g}{e} < 0$

Derivation: Algebraic expansion

Basis: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land d > 0 \land \frac{g}{e} < 0$, then
$(d + e x)^p (f + g x)^q == \left(-\frac{d^2 g}{e}\right)^q (d + e x)^{p-q} \left(1 + c^2 x^2\right)^q$

Rule: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land d > 0 \land \frac{g}{e} < 0$, then

$$\int (h x)^m (d + e x)^p (f + g x)^q (a + b \text{ArcSinh}[c x])^n \, dx \rightarrow \left(-\frac{d^2 g}{e}\right)^q \int (h x)^m (d + e x)^{p-q} \left(1 + c^2 x^2\right)^q (a + b \text{ArcSinh}[c x])^n \, dx$$

Program code:

```
Int[(h_.*x_)^m_.*(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^q*Int[(h*x)^m*(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0] && GtQ[d,0] && LtQ[g/e,0]
```

2: $\int (h x)^m (d + e x)^p (f + g x)^q (a + b \text{ArcSinh}[c x])^n \, dx$ when $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land \neg \left(d > 0 \land \frac{g}{e} < 0\right)$

Derivation: Piecewise constant extraction

Basis: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0$, then $\partial_x \frac{(d+e x)^q (f+g x)^q}{\left(1+c^2 x^2\right)^q} == 0$

Rule: If $e f + d g == 0 \land c^2 d^2 + e^2 == 0 \land (p \mid q) \in \mathbb{Z} + \frac{1}{2} \land p - q \geq 0 \land \neg \left(d > 0 \land \frac{g}{e} < 0\right)$, then

$$\int (h\,x)^m\,(d+e\,x)^p\,(f+g\,x)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{\left(-\frac{d^2 g}{e}\right)^{\text{IntPart}[q]}\,(d+e\,x)^{\text{FracPart}[q]}\,(f+g\,x)^{\text{FracPart}[q]}}{\left(1+c^2\,x^2\right)^{\text{FracPart}[q]}}\int (h\,x)^m\,(d+e\,x)^{p-q}\,\left(1+c^2\,x^2\right)^q\,(a+b\,\text{ArcSinh}[c\,x])^n\,dx$$

–

Program code:

```
Int[(h_.*x_)^m_.*(d_+e_.*x_)^p_*(f_+g_.*x_)^q_*(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  (-d^2*g/e)^IntPart[q]*(d+e*x)^FracPart[q]*(f+g*x)^FracPart[q]/(1+c^2*x^2)^FracPart[q]*
    Int[(h*x)^m*(d+e*x)^(p-q)*(1+c^2*x^2)^q*(a+b*ArcSinh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[e*f+d*g,0] && EqQ[c^2*d^2+e^2,0] && HalfIntegerQ[p,q] && GeQ[p-q,0]
```